
30. Loginis reiškinys. Loginis reiškinys gali įgyti dvi reikšmes: 0 arba 1. Kaip patikrinti, ar reiškinio reikšmė priklauso nuo kurio nors kintamojo (pavyzdžiui, a) reikšmės?

Imame visas galimas likusių kintamųjų reikšmių kombinacijas. Šiuo atveju turime keturis variantus:

	b
	c

	0
	0

	0
	1

	1
	0

	1
	1


Imame pirmąjį variantą (b = 0, c = 0) ir žiūrime, ar sutaps loginio reiškinio reikšmės, kai a = 0 ir a = 1. Jei ne, reiškinio reikšmė tikrai priklauso nuo kintamojo a. Jei sutampa, imame antrąją kintamųjų b ir c reikšmių kombinaciją ir vėl tikriname tą patį. Jei nors vieną kartą reikšmės nesutaps, tai rodys, kad reiškinio reikšmė tikrai priklauso nuo šio kintamojo.

Analogiškai nesunku patikrinti, ar reiškinio reikšmė priklauso nuo likusių kintamųjų b ir c.

Žemiau pateikiame funkciją šiam uždaviniui spręsti.

Kadangi kintamieji a, b ir c gali įgyti tik dvi reikšmes – 1 arba 0, tai funkcijoje jie aprašyti loginio tipo kintamaisiais.

type reiškinys = array [false..true, false..true, false..true] 

                                                    of boolean;

function priklauso (r: reiškinys): integer;

  var a, b, c, 

      nep_a, nep_b, nep_c: boolean; { nepriklauso }
begin
  { tikriname, ar reikšmė priklauso nuo kintamojo a }
  nep_a := true;

  for b := false to true do
    for c := false to true do
      nep_a := nep_a and (r[false, b, c] = r[true, b, c]);

  { tikriname, ar reikšmė priklauso nuo kintamojo b }
  nep_b := true;

  for a := false to true do
    for c := false to true do
      nep_b := nep_b and (r[a, false, c] = r[a, true, c]);

  { tikriname, ar reikšmė priklauso nuo kintamojo c }
  nep_c := true;

  for a := false to true do
    for b := false to true do
      nep_c := nep_c and (r[a, b, false] = r[a, b, true]);

  priklauso := ord(not nep_a) + ord(not nep_b) + ord(not nep_c)

end; { priklauso }
	Testo nr.
	Pradiniai duomenys
	Rezultatas
	Paaiškinimai

	1
	0 0 1 0 0 0 0 1
	3
	Priklauso nuo visų kintamųjų

	2
	1 0 1 0 0 1 0 1
	2
	Priklauso nuo kintamųjų a ir c

	3
	1 1 0 0 1 1 0 0
	1
	Priklauso nuo kintamojo b

	4
	0 0 0 0 0 0 0 0
	0
	Nepriklauso nė nuo vieno kintamojo


31. Reiškinys. Operacijas reikia parinkti keturiose vietose, iš viso yra galimos penkios operacijos. Taigi iš viso yra galimi 54 = 256 variantai. Tai yra labai nedaug, todėl uždavinį galima spręsti perrenkant visus galimus variantus. 

Patogiausia perrinkti naudojantis rekursija. Reiškinio reikšmę galima skaičiuoti įvairiais būdais. Svarbu neužmiršti operacijų prioritetų – pirma atliekama daugyba ir dalyba, po to –sudėtis ir atimtis. Daugybos ir dalybos operacijos atliekamos iš kairės į dešinę.

Aprašysime šitokį reiškinio skaičiavimo variantą. Panagrinėkime pavyzdį. Tarkime, kad kintamųjų reikšmės yra tokios: a ( 5, b ( 3, c ( 2, d ( 10, e ( 4, o nagrinėjamas operacijų rinkinys: +, *, /, -.

Kintamųjų reikšmes ir nagrinėjamą operacijų rinkinį saugosime dviejose lentelėse, pavyzdžiui:

	5
	3
	2
	10
	4


	+
	*
	/
	-
	


Pirmiausia atliekama daugyba, o likusių kintamųjų reikšmės bei operacijos perstumiamos po vieną langelį:

	5
	6
	10
	4
	


	+
	/
	-
	-
	


Tuomet atliekama dalyba:

	5
	0
	4
	
	


	+
	-
	
	
	


Galiausiai – sudėtis ir atimtis. Gauname, kad reiškinio reikšmė lygi 1. Gautą reikšmę belieka palyginti su kintamojo r reikšme. Sudėtį ir atimtį galima atlikti iš kairės į dešinę, todėl nebebūtina perstumti lentelės elementų.

const MAX = 5; { kintamųjų skaičius reiškinyje }
type kintamieji = array [1..MAX+1] of integer;

     operacijos = (sudėtis, atimtis, daugyba, liekana, dalmuo);

     rinkinys = array [1..max-1] of operacijos; { operacijų rinkinys }
procedure perstumti (i, liko: integer;

                     var sk: kintamieji;

                     var op: rinkinys);

{ pašalina i+1-ąjį kintamąjį bei i-osios operacijos ženklą }
  var j: integer;

begin
  for j := i to liko do
    begin
      sk[j + 1] := sk[j + 2];

      op[j] := op[j + 1]

    end
end; { perstumti }
procedure skaičiuoti (sk: kintamieji; op: rinkinys;

                      var rasta: boolean);

{ apskaičiuoja reiškinio reikšmę }
  var liko,  { tiek operacijų reikia atlikti }
      i, rez: integer;

      gerai: boolean; { ar neprireikė dalyti iš nulio }
begin
  liko := max - 1; { tiek operacijų reikia atlikti }
  { atliekamos daugybos ir dalybos operacijos }
  i := 1;

  gerai := true;

  while (i <= liko) and gerai do
    begin
      case op[i] of
        daugyba: if sk[i] < maxint div sk[i + 1]

                     then sk[i] := sk[i] * sk[i + 1]

                     else gerai := false;

        dalmuo: if sk[i + 1] = 0

                    then gerai := false

                    else sk[i] := sk[i] div sk[i + 1];

        liekana: if sk[i + 1] = 0

                     then gerai := false

                     else sk[i] := sk[i] mod sk[i + 1]

      end;

      if (op[i] in [daugyba, dalmuo, liekana]) and gerai

         then begin
                liko := liko - 1;

                perstumti (i, liko, sk, op)

              end
         else i := i + 1

    end;

  { atliekamos sudėties ir atimties operacijos }
  rez := sk[1];

  for i := 1 to liko do
    case op[i] of
      sudėtis: rez := rez + sk[i + 1];

      atimtis: rez := rez - sk[i + 1]

    end;

  rasta := gerai and (rez = sk[max + 1])

end; { skaičiuoti }
procedure parinkti (sk: kintamieji; n: integer;

                    var op: rinkinys;

                    var rasta: boolean);

{ parenka operaciją n-ajai pozicijai }
  var i: operacijos;

begin
  for i := sudėtis to dalmuo do
    if not rasta

       then begin
              op[n] := i;

              if n = max - 1

                 then skaičiuoti (sk, op, rasta)

                 else parinkti (sk, n + 1, op, rasta)

            end
end; { parinkti }
procedure ieškoti (sk: kintamieji;

                   var op: rinkinys;

                   var rasta: boolean);

{ ieško tinkamo operacijų rinkinio }
begin
  rasta := false;

  parinkti (sk, 1, op, rasta)

end; { ieškoti }
	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	1
	6 3 3 5 2 9
	6 + 3 + 3 - 5 + 2 ( 9 
	Reiškinyje yra tik sudėties ir atimties operacijos

	2
	3 4 4 57 32 23
	3 * 4 * 4 - 57 + 32 = 23
	Reiškinyje yra daugybos operacijų; operacijų atlikimo tvarka nesvarbi

	3
	13 3 6 14 2 23
	13 + 3 \ 6 + 14 / 2 = 23
	Reiškinyje yra daugybos operacijų, operacijų atlikimo tvarka svarbi

	4
	3 16 3 5 4 5
	3 + 16 * 3 / 5 / 4 = 5
	Dalybos operacija negali būti atlikta pirmiau nei daugybos

	5
	10 2 2 2 1 1
	10 - 2 * 2 * 2 - 1 ( 1
	Bandant visus variantus galima dalyba iš nulio

	6
	2 10 2 1 3 3000
	Sprendinio nėra
	Sprendinio nėra


32. Dvejetainis kodas. Sprendimo idėja paprasta. Jei skaičiai a ir b sutampa, tai reiškia, kad nė vienas laidas nenutrūkęs. Uždavinys išspręstas.

Jei a ir b nesutampa, juos paverčiame dvejetainiais skaičiais ir įsidėmime lentelėje (vienmačiame masyve). Tuomet pakanka palyginti lenteles pradedant dešiniausia skiltimi. Jei jos skiriasi tik vienu skaitmeniu ir tas skaitmuo dvejetainiame skaičiaus a užraše lygus 1, o dvejetainiame b užraše – 0, tuomet hipotezė teisinga. Bet kuriuo kitu atveju hipotezė klaidinga.

Atskirai panagrinėkime, kaip dešimtainis skaičius perrašomas dvejetainiu. Dvejetainio skaičiaus skaitmenys gaunami dalijant duotą skaičių iš dvejeto (dvejetainės sistemos pagrindo). Tai – seka dalybos liekanų, išdėstytų atvirkščia tvarka negu dvejetainio skaičiaus skaitmenys. Pavyzdžiui, panagrinėkime skaičiaus 23 užrašymą dvejetainiu (32.1 pav.).

32.1 pav.

Šio skaičiaus dalybos iš dvejeto liekanos yra 11101. Jas išdėstę atvirkščia tvarka, gauname 10111. Tai ir bus skaičiaus 23 dvejetainis užrašas:

2310 = 24 ( 1 ( 23 ( 0 + 22 ( 1 + 21 ( 1 + 20 ( 1 = 101112
const N = ...;   { maksimalus dvejetainio skaičiaus ilgis }
type dvejetainis = array [1..N] of 0..1;

procedure dvejet (a: integer; var dv: dvejetainis);

{ skaičius a užrašomas dvejetainiu }
  var i, j: integer;

      tarp: dvejetainis;

begin
  i := 1;

  while a div 2 <> 0 do    { įsimenamos dalybos iš 2 liekanos }
    begin
      tarp[i] := a mod 2;

      a := a div 2;

      i := i + 1

    end;

  tarp[i] := a mod 2;

  for j := 1 to N do
    dv[j] := 0;

  { gautos liekanos užrašomos atvirkščia tvarka }
  for j := 1 to i do
    dv[N -j + 1] := tarp[j]

end; { dvejet }
Spręsdami šį uždavinį laikome, kad pradiniai duomenys yra neneigiami skaičiai. Dar reikėtų aptarti, kokio didumo turėtų būti lentelė dvejetainiams skaičiams įsiminti. Užtektų 15 skilčių. Tiek skaitmenų turi didžiausio leistino sveikojo skaičiaus maxint dvejetainis užrašas.

const N = 15;   { maksimalus dvejetainio skaičiaus ilgis }
type dvejetainis = array [1..N] of 0..1;

procedure hipotezė (a, b: integer;

                    var teisinga: boolean; { ar teisinga hipotezė }
                    var laidas: integer { nutrūkusio laido numeris });

  var dva, dvb: dvejetainis; { a ir b užrašyti dvejetaine sistema }
      i: integer;

begin
  laidas := 0;

  teisinga := true;

  if a <> b then { jei a ir b nesutampa }
   begin
     dvejet (a, dva); { jie perrašomi dvejetainiu pavidalu }
     dvejet (b, dvb);

     i := N;

     while teisinga and (i >= 1) do { ieškomas nutrūkęs laidas }
       begin
         if dva[i] <> dvb[i] 

         then if (dva[i]=1) and (dvb[i]=0) and (laidas=0)

              then laidas := N - i + 1

              else teisinga := false;

         i := i - 1

       end
   end
end; { hipotezė }
	Testo nr.
	Pradiniai duomenys
	Rezultatas
	Paaiškinimai

	1
	542 542
	SKAIČIUS PERDUOTAS TEISINGAI
	

	2
	53 55
	HIPOTEZĖ KLAIDINGA
	5310 = 1101012, 

5510 = 1101112,

Skaičiai skiriasi tik vienu skaitmeniu, tačiau hipotezė klaidinga, nes nulis pavirto vienetu, o ne atvirkščiai

	3
	476 332
	HIPOTEZĖ KLAIDINGA
	47610 = 1110111002, 

33210 = 1010011002,

Du nuliai virto vienetais

	4
	65 1
	HIPOTEZĖ TEISINGA

NUTRŪKUSIO LAIDO NUMERIS:7
	6510 = 10000012, 

110 = 12,

Užrašyti dvejetainiu pavidalu skaičiai turi skirtingą skaitmenų kiekį

	5
	8790 7766
	HIPOTEZĖ KLAIDINGA
	879010 = 100010010101102, 

776610 = 11110010101102;

Pradiniai duomenys skiriasi dvejeto laipsniu

	6
	360 296
	HIPOTEZĖ TEISINGA

NUTRŪKUSIO LAIDO NUMERIS:7
	36010 = 1011010002, 

29610 = 1001010002,

Ribinis atvejis 

	7
	32767 31743
	HIPOTEZĖ TEISINGA

NUTRŪKUSIO LAIDO NUMERIS:11
	3276710 = 1111111111111112, 

3174310 = 1111011111111112,

Ribinis atvejis 


33. Kubinė šaknis. Kadangi sąlygoje kalbama apie sveikuosius skaičius, reikėtų sudaryti kubinės šaknies traukimo funkciją sveikiesiems skaičiams. Realiųjų skaičių funkcija čia netinka, nes galime gauti netikslų rezultatą.

Tačiau esama kur kas geresnio algoritmo: galima tirti sveikųjų skaičių kubus. Didžiausias skaičius, kurio skaitmenys skirtingi, turi 10 skaitmenų. Jis lygus 9876543210, t. y. jis viršija maxlongint, todėl teks modeliuoti veiksmus su dideliais skaičiais. 

Pasirenkame skaičių, kurio kubas turi daugiau nei 10 skaitmenų (pavyzdžiui, 3000). Šį skaičių pakeliame kubu, patikriname, ar kubą sudarantys skaitmenys yra skirtingi. Jei taip – gavome ieškomą kubą, jei ne – skaičių sumažiname vienetu ir kartojame veiksmus, kol negausime rezultato. 

Atskirai panagrinėsime, kaip skaičius keliamas kubu. Pirmiausia jį pakelsime kvadratu (kvadratas tikrai neviršys maxlongint) ir turimą kvadratą užrašysime į masyvą (t. y. kvadrato reikšmę sumodeliuosime masyvu). Tuomet modeliuosime skaičiaus, vaizduojamo masyvu, sandaugą iš sveikojo tipo skaičiaus – turimą kvadratą vis dauginsime iš atitinkamo skaitmens ir pastūmę į kairę pridėsime prie rezultato (33.1 pav.).

33.1 pav.

const ilgis   = 11;

type  didelis = array [1..ilgis] of 0..9; { didelis skaičius }
procedure kubas (var sk: didelis );

  var s: longint;

      rasta: boolean;

  procedure kelti_kubu (s: longint; var sk: didelis );

  { skaičius s keliamas kubu }
    var kv: longint;

        perk,      { perkėlimas }
        post,      { postūmis nuo skaičiaus pradžios dauginant }
        skaitmuo, i, j, tarp: integer;

        k: didelis; { vienas iš daugiklių – pradinio skaičiaus kvadratas }
  begin
    kv := s * s;  { skaičių pakeliame kvadratu }
    i := ilgis;

    for i := ilgis downto 1 do   { jį perrašome masyvu }
      begin
        k[i] := kv mod 10;

        kv := kv div 10

      end;

    for i := ilgis downto 1 do  { čia bus talpinamas rezultatas }
      sk[i] := 0;

    post := Ilgis;

    while s <> 0 do     { dauginama iš skaičiaus s kiekvieno skaitmens }
      begin
        skaitmuo := s mod 10;

        perk := 0;

        for j := post downto 1 do
         begin
           tarp := k[Ilgis-(post-j)] * skaitmuo + perk + sk[j];

           sk[j] := tarp mod 10;

           perk := tarp div 10

         end;

        post := post - 1;

        s := s div 10

      end
  end; { kubas }
  function skirtingi (sk: didelis): boolean;

  { ar visi natūraliojo skaičiaus sk skaitmenys skirtingi }
    var s: set of 0..9;  { skaičiaus skaitmenų aibė }
        skirt: boolean;

        i: integer;

  begin
    i := 1;       { praleidžiame nulius skaičiaus priekyje }
    while sk[i] = 0 do
      i := i + 1;

    s := [ ];  skirt := true;

    while (i <= Ilgis) and skirt do
      begin
        if sk[i] in s  { ar skaitmuo yra aibėje s }
           then skirt := false

           else s := s + [sk[i]];

        i := i + 1

      end;

    skirtingi := skirt

  end;

begin
  s := 3000;      { pasirenkamas skaičius, kurio kubas viršija 10 skaitmenų }
  rasta := false;

  while not rasta do
    begin
      s := s - 1;

      kelti_kubu (s, sk);

      rasta := skirtingi (sk)

    end
end; { skirtingi }
	Rezultatas

	32461759


34. Klaida renkant skaičius. Pažymėkime duotuosius skaičius a, b ir c. Galimi tokie atvejai:

1. Klaidos nebuvo padaryta; tada turime tik vieną sprendinį: a, b, c.

2. Klaida padaryta skaičiuje a; gauname a’ = c - b, sprendinys a’, b, c.

3. Klaida padaryta skaičiuje b; gauname b’ = c - a, sprendinys a, b’, c.

4. Klaida padaryta skaičiuje c; gauname c’ = a + b, sprendinys a, b, c’.

Jei nė vienas aukščiau išvardytų atvejų netinka, tada skaičių atstatyti negalima. Daugiausia gali tikti trys atvejai – toks yra maksimalus sprendinių skaičius.

Patogu parašyti atskiras funkcijas, kurios tikrintų, ar renkant skaičių nebuvo padaryta kuri nors klaida. 

Viena funkcija tikrins, ar renkant skaičių x, nebuvo paspaustas kitas klavišas (t. y. surinktas neteisingas skaitmuo) ir gautas skaičius y. Tikrinama paprastai: pradedant nuo galo sulyginami atitinkami abiejų skaičių skaitmenys. Klaida buvo padaryta, jei skaičiai skiriasi tik vienu skaitmeniu.

Kita funkcija tikrins, ar renkant skaičių x nebuvo praleistas skaitmuo ir gautas skaičius y. Pirmiausia atmetami vienodi abiejų skaičių skaitmenys, pradedant nuo paskutiniųjų. Tuomet atmetamas vienas skaičiaus x skaitmuo ir tikrinama, ar x = y.

const Max = 3;  { maksimalus sprendinių skaičius }
type rezult = array [1..Max] of record
                                  a, b, c: integer

                                end;

procedure taisyti (a, b, c: integer;  { pradiniai skaičiai }
                   var sk: integer;   { sprendinių skaičius }
                   var rez: rezult);  { ir patys sprendiniai }
  var teis: integer;

  procedure įrašyti (a, b, c: integer);

     { skaičiai a, b, c įrašomi į rezultatų masyvą rez }
     { rez ir sk – globalieji kintamieji }
  begin
    sk := sk + 1;

    rez[sk].a := a;

    rez[sk].b := b;

    rez[sk].c := c

  end; { įrašyti }
  function praleistas (x, y: integer): boolean;

  { tikrinama, ar renkant x, nebuvo praleistas skaitmuo ir gautas skaičius y }
  begin
    { praleidžiami vienodi skaitmenys }
    while (x mod 10 = y mod 10) and (x > 0) do
      begin
        x := x div 10;

        y := y div 10

      end;

    x := x div 10;

    praleistas := x = y

  end; { praleistas }
  function klaida (x, y: integer): boolean;

  { tikrinama, ar renkant nebuvo padaryta klaida – gal x ir y skiriasi tik vienu skaitmeniu }
    var kl_sk: integer; { keliais skaitmenimis skiriasi skaičiai }
  begin
    kl_sk := 0;

    while ((x <> 0) or (y <> 0)) and (kl_sk <= 1) do
       begin
         if x mod 10 <> y mod 10

            then kl_sk := kl_sk + 1;

         x := x div 10;

         y := y div 10

        end;

     klaida := kl_sk = 1

  end; { klaida }
begin
  sk := 0;

  if a + b = c 

  then įrašyti (a, b, c)

  else 

    begin
      { ieškome klaidos skaičiuje a }
      teis := c - b;

      if praleistas (teis, a) or praleistas (a, teis) or 

         klaida (teis, a)

         then įrašyti (teis, b, c);

      { ieškome klaidos skaičiuje b }
      teis := c - a;

      if praleistas (teis, b) or praleistas (b, teis) or
         klaida (teis, b)

         then įrašyti (a, teis, c);

      { ieškome klaidos skaičiuje c }
      teis := a + b;

      if praleistas (teis, c) or praleistas (c, teis) or
         klaida (teis, c)

         then įrašyti (a, b, teis)

    end 

end; { taisyti }
	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	1
	328 63 391
	1

328 63 391
	Klaidos nebuvo padaryta

	2
	103 213 1516
	1

1303 213 1516
	Skaičiuje a praleistas skaitmuo

	3
	32 4800 512
	1

32 480 512
	Skaičiuje b įterptas skaitmuo

	4
	32 64 16
	1

32 64 96
	Skaičiuje c skaitmuo surinktas klaidingai

	5
	13886 6119 5
	1

13886 6119 20005
	Skaičiuje c praleistas pirmasis skaitmuo, tuomet nebeliko reikšminių nulių

	6
	3582 111 693
	2

582 111 693

3582 111 3693
	Galimi du sprendiniai

	7
	1234 4321 5655
	3

1334 4321 5655

1234 4421 5655

1234 4321 5555
	Galimi trys sprendiniai

	8
	4058 369 7569
	0
	Skaičių atstatyti neįmanoma


35. Skaitmenų suma ir sandauga. Iš pirmo žvilgsnio atrodo, kad reikėtų sandaugą skaidyti dauginamaisiais ir tada iš gautų skaitmenų sudarinėti skaičių, tenkinantį sąlygą. Tačiau sandauga gali būti išskaidoma skaitmenimis daugeliu būdų ir gana keblu iš jų sudaryti mažiausią skaičių.

Paprastesnis sprendimas būtų generuoti visus skaičius (didėjimo tvarka), kurių skaitmenys išdėstyti iš kairės į dešinę nemažėjančiai (taip išvengtume skaičių su tais pačiais skaitmenimis), pavyzdžiui, iš eilės eitų 298, 299, 333, 334 ir t. t. Tuomet bereikėtų patikrinti, ar skaičius tenkina sąlygą, t. y. ar jo skaitmenų suma ir sandauga lygios atitinkamiems pradiniams duomenims. Tokių skaičių generavimui sudaroma procedūra naujas. Beje, nesunkiai galima būtų padaryti procedūrą, kuri didėjimo tvarka iš karto generuotų skaičius, kurių skaitmenų suma lygi duotam pradiniam duomeniui. Tai padaryti paliekame skaitytojams.

Kiekvieną tikrinamą skaičių patogiausia laikyti masyve, sudarytame iš skaičiaus skaitmenų. Tuomet nebereikia papildomų veiksmų skaičiaus skaidymui skaitmenimis, o svarbiausia – skaičius gali būti didelis. Iš tikrųjų, sandauga gali būti artimas maxint skaičius, tuomet ieškomasis skaičius gali turėti iki 15 skaitmenų (215 - 1 = maxint). 

Reikia išskirti tuos atvejus, kai sandaugos neįmanoma išskaidyti skaitmenimis, pavyzdžiui, 121 (tai pirminių skaičių 11 ( 11 sandauga), tuomet nėra prasmės toliau ko nors beieškoti.

Dar vienas keblumas kyla, kai randame skaičių, kurio skaitmenų sandauga tenkina sąlygą, tačiau duota skaitmenų suma yra gerokai didesnė. Tokiu atveju ieškomasis skaičius gali būti sudarytas tik šitaip: prie rastojo mažiausio (pirmojo) skaičiaus, kurio skaitmenų sandauga tenkina sąlygą, priekyje reikia prirašyti tiek vienetų, kol pasidarys reikalinga skaitmenų suma. Vienetus prirašome prie mažiausio rastojo skaičiaus. 

Yra dar vienas atvejis – kai sandauga lygi nuliui. Tokiu atveju visada rasime skaičių, kurio skaitmenų suma bus lygi duotajai. Tuomet prieš pirmą skaitmenį teliks įterpti nulį.

Algoritmą išreiškiančioje procedūroje – keturi rezultatai. Loginis kintamasis rasta rodo, ar sprendinys rastas. Jei taip, tai rastasis skaičius įrašytas masyve sk. Jei ne, tuomet rezultatas dar gali būti (kai surasta tenkinanti sąlygą skaitmenų sandauga) sudarytas iš dviejų dalių: masyve dalis esančio skaičiaus ir prieš jį prirašomų vienetų (jų skaičius lygus kintamojo vienetai reikšmei).

const MAX = 15; { kad tilptų integer tipo išskaidyta skaitmenimis sandauga }
type skaitmuo = 0..9;

     skaičius = array [0..MAX] of skaitmuo;

procedure naujas (var sk: skaičius; var galima: boolean);

{ randa naują skaičių, didesnį už prieš buvusį ir nemažėjančia tvarka išdėstytais skaitmenimis } 
  var skm, i: integer;

begin
  galima := true;

  skm := MAX;

  while (sk[skm] = 9) and (skm <> 0) do
     skm := skm - 1;

  { skaitmenų sumos minimumas }
  if skm <> 0

     then begin
            sk[skm] := sk[skm] + 1; { didiname atitinkamą skaitmenį }
            for i := skm + 1 to MAX do     { sulyginame į dešinę }
              sk[i] := sk[skm]             { esančius skaitmenis }
          end
      else galima := false

end; { naujas }
procedure mažiausias_skaičius (sum, san: integer;

                               var sk: skaičius; 

                               var rasta: boolean;

                               var dalis: skaičius; 

                               var vienetai: integer);

{ jei skaičiaus nerasta, bet jo sandauga išskaidoma skaitmenimis, }
{ o skaitmenų suma gana didelė, tuomet ieškomasis skaičius bus sudarytas }
{ iš rasto skaičiaus dalies ir priekyje esančių vienetų }
  var ssum, ssan, i: integer;

      galima, pirmas: boolean;

begin
  { tikriname, ar duotą sandaugą galima išskaidyti skaitmenų sandauga; }
  { jei negalima, toliau nėra prasmės ieškoti norimo rezultato }
  ssan := san; i := 2;

  while (i < 9) and (ssan > 9) do
    if ssan mod i = 0

       then ssan := ssan div i

       else i := i + 1;

  rasta := ssan <= 9;

{ ieškome skaičiaus } 

for i := 0 to MAX do
    sk[i] := 0;

  sk[max] := 1; { generuojame pirmąjį skaičių }
  ssum := 1; ssan := 1;

  vienetai := 0;

  pirmas := true; { reikalingas tik pirmą kartą rastas skaičius, }
                  { kurio skaitmenų sandauga tenkina sąlygą }
  galima := true; { galima sudaryti didesnį skaičių }
  while galima and ((ssum<>sum) or (ssan<>san) and (san<>0)) do
    begin
      naujas (sk, galima);

      { rasime sk skaitmenų sumą ir sandaugą }
      ssum := 0; ssan := 1;

      for i := 1 to max do
        begin
          ssum := ssum + sk[i];

          if sk[i] <> 0

             then if (ssan <= maxint div sk[i])

                     then ssan := ssan * sk[i]

                     else ssan := 0

        end;

      { pasižymime pirmąjį skaičių, kurio skaitmenų sandauga }
      { tenkina sąlyga, o skaitmenų suma – ne (yra mažesnė) }
      if (ssan = san) and (ssum < sum) and pirmas

          then begin
                 dalis := sk;

                 vienetai := sum - ssum;

                 pirmas := false

                end
    end;

  if san = 0 { įterpiame nulį prieš pirmą skaičių }
     then begin
            i := 1;

            while sk[i] = 0 do
              i := i + 1;

            sk[i - 1] := sk[i];

            sk[i] := 0;

          end;

  rasta := (ssum = sum) and (ssan = san) or (san = 0)

end; { mažiausias_skaičius }
	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	1
	5 5
	5
	Kai suma ir sandauga yra sutampantys vienženkliai skaičiai, laikome, kad rezultatas irgi vienženklis skaičius

	2
	8 15
	35
	Paprastas duomuo, kai sandauga išsiskaido į du dauginamuosius: 3 ir 5

	3
	7 12
	34
	Sandaugą galima išskaidyti keliais būdais: 2, 2, 3, 3 ir 4, 2 ir 6

	4
	22 360
	598
	Sandaugą galima išskaidyti daugeliu būdų

	5
	27 19683
	333333333
	Gaunamas didelis skaičius

	6
	7 169
	Sprendinio nėra
	Sandaugos negalima išskaidyti skaitmenimis

	7
	10 6
	11116
	Skaitmenų suma šiek tiek didesnė už sandaugą

	8
	25 6
	11111111111111111116
	Didelė skaitmenų suma: skaičiaus priekyje prirašomi trūkstami vienetai

	9
	12 0
	309
	Sandauga lygi 0


36. Paskalio kalbos skaičius. Norint nustatyti, ar duota simbolių eilutė yra Paskalio programos skaičius, pirmiausia reikia išnagrinėti skaičių aprašančias taisykles. Vaizdžiausiai šios taisyklės išreiškiamos sintaksės diagramomis (36.1 pav.). 

36.1 pav.

Atkreipiame dėmesį: jei nėra kitokių nurodymų, visuomet nagrinėjame Paskalio kalbos standartą.

Sintaksės diagramomis griežtai nusakoma, kas leistina skaičiuje ir kas neleistina. Todėl pagal jas nesunku parašyti programą, kuri patikrintų, ar duota simbolių eilutė yra Paskalio kalbos skaičius, ar ne. Diagramų veiksmus geriausia išreikšti atskiromis procedūromis. 

Turbo Paskalyje yra standartinė procedūra val (eilutė, reikšmė, kodas), kuri apskaičiuoja simbolių eilutės reikšmę. Jei tai padaryti galima, tuomet kodas=0, priešingu atveju kodas rodo netinkamo simbolio numerį. Vartojant šią procedūrą, visiškai supaprastėja programa. Tačiau šia procedūra nereikėtų labai žavėtis. Pirmiausia, ji turi ribojimų: jei skaičiaus (simbolių eilutės) gale yra tarpų, tai toks skaičius nelaikomas Paskalio kalbos skaičiumi. Toliau, jos traktuojamo Paskalio skaičiaus samprata nesutampa su Paskalio kalbos standartu, pavyzdžiui, skaičiumi laikoma 5. arba .5 (taškas po skaičiaus ar prieš jį). Todėl geriau pačiam sudaryti algoritmą, o ne naudotis netikslia standartine procedūra.

type atsakymas = record
                   sveikas,                { sveikasis skaičius }
                   realus,                 { realusis skaičius }
                   ne_skaičius: boolean    { ne Paskalio skaičius }
                 end;

  { žemiau išvardinti globalieji kintamieji }
  var eil: string;     { pradinis duomuo }
      tipas: atsakymas; { rezultatas }
      reali: real;          { realaus skaičiaus reikšmė  (rezultatas) }
      sveika,               { sveikojo skaičiaus reikšmė  (rezultatas) }
      ilgis, i: integer;               { eilutės ilgis ir indeksas }
procedure tarpai;

{ praleidžiami tarpai eilutės pradžioje ir gale }
   var j: integer;

begin
  ilgis := length (eil);

  i := 1; { praleidžiame pradžioje esančius tarpus }
  while (eil[i] = ' ') and (i < ilgis) do
    i := i + 1;  { i rodo pirmą ne tarpo simbolį }
  j := ilgis;

  while (eil[j] = ' ') and (j > 1) do
    j := j - 1;  { išmetame tarpus skaičiaus gale }
  ilgis := j     { eilutės ilgis atmetus tarpus gale }
end; { tarpai }
procedure skaitmenų_seka;

{ eilutėje praleidžiama ištisinė skaitmenų seka }

begin
  if eil[i] in ['0'.. '9']

    then { turi būti bent vienas skaitmuo }
       while (eil[i] in ['0'..'9']) do
          i := i + 1  { i rodys į pirmą ne skaitmenį }
    else i := 0  { skaitmenų seka tuščia }
end; { skaitmenų_seka }
procedure laipsnio_rodiklis;

begin
  if eil[i] in [ '-', '+'] then i := i + 1;

  skaitmenų_seka

end; { laipsnio rodiklis }
function sv_reikšmė: integer;

{ skaičiuojama Paskalio skaičiaus reikšmė }
{ eilutės eil dalį nuo i-tojo elemento paverčia sveikuoju skaičiumi }
  var rez: integer;

      neig: boolean;

begin
  rez := 0;

  neig := eil[i] = '-';

  if eil[i] in ['-', '+']

     then i := i + 1;

  while eil[i] in ['0'.. '9'] do
    begin
      rez := rez * 10 + ord (eil[i]) - ord ('0');

      i := i + 1

    end;

  if neig then rez := rez * (-1);

  sv_reikšmė := rez

end; { sv_reikšmė }
procedure rezultatas;

{ apskaičiuoja skaičiaus reikšmę }
  var daug: real;  { realiojo skaičiaus daugiklis }
      laip: string;

      laipsnis, j: integer;

begin
  { praleisime tarpus pradžioje }
  i := 1;

  while eil[i] = ' ' do
    i := i + 1;

  if tipas.sveikas { sveikojo skaičiaus reikšmė }
  then sveika := sv_reikšmė

  else 

   begin   { realiojo skaičiaus reikšmė }
     reali := sv_reikšmė;

     if eil[i] = '.'    { prijungiame skaitmenis po kablelio }
     then begin
            i := i + 1;

            daug := 1;

            while eil[i] in ['0'..'9'] do
              begin
                daug := daug/10;

                reali := reali + (ord(eil[i])-ord ('0'))*daug;

                i := i + 1

              end
          end;

     if eil[i] in ['e', 'E'] { dauginame iš dešimties laipsnio }
     then begin
            i := i + 1;

            laip := copy (eil, i, ilgis-i-1);

            laipsnis := sv_reikšmė;

            for j := 1 to abs (laipsnis) do
              if laipsnis < 0

                 then reali := reali / 10

                 else reali := reali * 10

          end
    end
end; { rezultatas }
procedure skaičius;

begin  { skaičius }
  tarpai; { praleidžiami tarpai iš abiejų eilutės pusių }
  eil := eil + '='; { išvengsime tikrinimo, ar nepasibaigė eilutė }
  ilgis := ilgis + 1;

  if (eil[i] = '-') or (eil[i] = '+') then i := i + 1;

  { ar tai sveikasis skaičius? }
       skaitmenų_seka;

       tipas.sveikas := i = ilgis;

       tipas.ne_skaičius := not tipas.sveikas;

       { jei pasiekėme eilutės galą, vadinasi, tai būta sveikojo skaičiaus }
       { jei ne – ... }
  { gal tai realusis skaičius? }
       if not tipas.sveikas and (eil[i] in ['.', 'E', 'e'])

         then begin
                if eil[i] = '.'

                  then begin i := i + 1;

                             skaitmenų_seka

                       end;

                if (eil[i] = 'E') or (eil[i] = 'e')

                  then begin i := i + 1;

                             laipsnio_rodiklis

                       end;

                tipas.realus := i = ilgis;

                tipas.ne_skaičius := not tipas.realus

              end;

  if not tipas.ne_skaičius then rezultatas

end; { skaičius }
	Testo nr.
	Pradiniai duomenys
	Rezultatas
	Paaiškinimai

	1
	255
	SVEIKASIS 255
	Sveikasis natūralusis skaičius

	2
	-25
	SVEIKASIS -25
	Sveikasis neigiamas skaičius

	3
	25.5
	REALUSIS 25.5
	Paprastas realusis skaičius

	4
	7e2
	REALUSIS 7.0000000000E+02
	Realusis su laipsnio rodikliu

	5
	-7E3
	REALUSIS -7.0000000000E+03
	Neigiamas realusis, didžioji E

	6
	-7e-2
	REALUSIS -7.0000000000E-02
	Neigiamas laipsnio rodiklis

	7
	+22.5e-13
	REALUSIS 2.2500000000E-12
	Sudėtingas realusis skaičius

	8
	5.
	NE PASKALIO SKAIČIUS
	Realusis skaičius negali baigtis tašku

	9
	.5
	NE PASKALIO SKAIČIUS
	Realusis skaičius negali prasidėti tašku

	10
	2,5
	NE PASKALIO SKAIČIUS
	Realiojo skaičiaus užraše negalima vartoti kablelio

	11
	e8
	NE PASKALIO SKAIČIUS
	Realųjį skaičių negali sudaryti tik laipsnio rodiklis

	12
	2e1.2
	NE PASKALIO SKAIČIUS
	Laipsnio rodiklis negali būti trupmeninis.

	13
	2e 1
	NE PASKALIO SKAIČIUS
	Skaičiuje tarpai neleistini


37. Būdvardžio linksniuotė. Būdvardžiai turi tris linksniuotes: pirmąją, antrąją ir trečiąją. Jos nustatomos pagal būdvardžio vienaskaitos vardininko galūnę. Taigi daugiskaitos vardininkas čia nereikalingas, nors dauguma gramatikų teberašo, kad būdvardžių linksniuotės nustatomos iš vienaskaitos ir daugiskaitos vardininkų (likęs nuo kažkurių laikų rudimentas): galbūt yra vienas kitas žodis – išimtis, tam reikėtų rimtesnių kalbos studijų.

	Linksniuotės

(vienaskaitos vardininko galūnės)

	I
	II
	III

	vyr. g.
	mot. g.
	vyr. g.
	mot. g.
	vyr. g.
	mot. g.

	-as
	-a
	-us
	-i
	-is

-ys
	-ė


Pavyzdžiai

	I
	II
	III

	baltas, balta
	atsargus, atsargi
	medinis, medinė

	šviežias, šviežia
	imlus, imli
	baltesnis, baltesnė

	gražiausias, gražiausia
	
	kairys, kairė


const n = 32;  { ilgiausias žodis }
type linksniuotė = (pirma, antra, trečia);

     žodis = string[n];

function kuri_linksniuotė (žod: žodis): linksniuotė;

{ būdvardžio linksniuotės nustatymas }

  var žod_ilgis: integer;

begin

  žod_ilgis := length (žod);

  if žod[žod_ilgis] in ['s', 'S']

     then       { vyriškoji giminė }
       case žod[žod_ilgis-1] of
         'a', 'A': kuri_linksniuotė := pirma;

         'u', 'U': kuri_linksniuotė := antra;

         'i', 'I', 'y', 'Y': kuri_linksniuotė := trečia

       end
     else       { moteriškoji giminė }
       case žod[žod_ilgis] of
           'a', 'A': kuri_linksniuotė := pirma;

           'i', 'I': kuri_linksniuotė := antra;

           'ė', 'Ė': kuri_linksniuotė := trečia

       end
end; { kuri_linksniuotė }
	Testo nr.
	Pradiniai duomenys
	Rezultatas

	1
	VARINIS VARINIAI
	TREČIA

	2
	KAIRYS KAIRIAI
	TREČIA

	3
	GRAŽIAUSIAS GRAŽIAUSI
	PIRMA

	4
	ATSARGUS ATSARGŪS
	ANTRA

	5
	IMLI IMLŪS
	ANTRA

	6
	ŠVIEŽIA ŠVIEŽIOS
	PIRMA

	7
	BALTESNĖ BALTESNĖS
	TREČIA


Pastaba. Nors pradiniuose duomenyse pateikėme ir daugiskaitos vardininką, tačiau algoritme tuo nesinaudojama. Todėl funkcijos pradinis duomuo yra vienas žodis – vienaskaitos vardininkas.

II etapas

38. Riestainiai. Jei yra n riestainių ir juos reikia padalinti m vaikų, tai aišku, kad vienas vaikas turi gauti n/m visų riestainių. Vadinasi, pirmasis sąlygos reikalavimas aiškus. 

Paprasčiausias pjaustymo būdas, tenkinantis antrąjį reikalavimą (kad riestainiai būtų pjaustomi į lygias dalis), būtų kiekvieną riestainį supjaustyti į m dalių. Kiekvienas vaikas gautų po vieną kiekvieno supjaustyto riestainio dalį. Iš viso jis gautų m dalių. Toks pjaustymas atitiktų trupmenos skaidymą į n dėmenų:

n/m = 1/m + 1/m + 1/m + … +1/m
                             n
Tarkime, kad turime 5 riestainius, kuriuos reikia padalyti 6 vaikams. Kiekvieną riestainį supjaustykime į 6 dalis ir kiekvienam vaikui duokime po vieną kiekvieno riestainio dalį (38.2 pav.):

38.2 pav.

5/6 ( 1/6 + 1/6 + 1/6 + 1/6 +1/6;

O dabar tarkime, kad trys riestainiai pjaustomi į dvi dalis, o du riestainiai į tris dalis. Iš viso yra 6 didesnės ir 6 mažesnės riestainio dalys. Kiekvienas vaikas gaus po vieną didesnę ir po vieną mažesnę dalį:

5/6 ( 1/2 + 1/3 ( 3/6 + 2/6

Matome, kad kiekvienas dėmuo reiškia riestainio pjaustymą (1/2 reiškia, kad riestainis buvo perpjautas į dvi dalis). Kuo didesnis vardiklis, į tuo daugiau dalių buvo pjaustytas riestainis. 

Norint, kad pjaustymų skaičius būtų mažesnis, reikia stengtis trupmeną išskaidyti į kiek galint mažiau dėmenų, turinčių pavidalą 1/k, kai k yra m daliklis. Taigi visas uždavinio sprendimas pavirsta trupmenos n/m skaidymu.

program riestainiai;

   var f: text;       { pradinių duomenų byla }
       n, m,          { riestainių ir vaikų skaičius }
       k,             { į kiek dalių pjaustyti: m mod k = 0 }
       r, d: integer;

begin
  { pradinių duomenų skaitymas }
  assign (f, 'DUOMENYS.DAT');

  reset(f);

  readln (f, n, m);

  close (f);

  { riešutų dalybos }
  writeln('RIESTAINIŲ     Į KIEK DALIŲ   KIEK DALIŲ DUOTI');

  writeln(' KIEKIS         PJAUSTYTI     KIEKVIENAM VAIKUI');

  for k := 1 to m do
    if m mod k = 0 then   { kiekvienas riestainis pjaustomas į lygias dalis }
      begin
        r := m div k;   { 1/k = r/m, tad r = m/k, riestainių kiekis, kad }
                            { kiekvienam vaikui užtektų po vieną gabalą }
        d := n div r;                      { po kiek dalių duoti }
        n := n mod r; { n/m := n/m - d/k }               { kiek liko }
        if d > 0 then
          writeln (r * d: 6, k: 17, d: 24)

      end
end.

	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	
	
	RIESTAINIŲ

KIEKIS
	Į KIEK DALIŲ

PJAUSTYTI
	KIEK DALIŲ DUOTI

KIEKVIENAM VAIKUI
	

	1
	5 6
	3

2
	2

3
	1

1
	Sąlygoje buvęs pavyzdys

	2
	5 7
	5
	7
	5
	Išskaidyti negalima, kiekvieną riestainį teks pjaustyti į 7 dalis, supjausčius gausime 35 dalis, kiekvienas vaikas gauna po 5 dalis

	3
	6 12
	6
	2
	1
	Trupmeną iš karto galima suprastinti ir toliau skaidyti nereikia;

	4
	9 12
	6

3
	2

4
	1

1
	Trupmeną tenka ir skaidyti ir prastinti

	5
	10 8
	8

2
	1

4
	1

1
	Riestainių daugiau nei vaikų


39. Perpieštas ornamentas. Jei ornamentas būtų perpieštas lapo viduryje, būtų nesunku patikrinti, ar jis perpieštas teisingai. Tokiu atveju būtų aišku, kur yra simetrijos ašis. Sunkiau, kai ašis nežinoma. 

Pirmas algoritmo žingsnis – surasti mažiausią stačiakampį, į kurį telpa ornamentas.

Išskirkime keletą atvejų:

a) stačiakampis perpieštas teisingai;

b) padaryta viena klaida mažiausiame stačiakampyje, kuriame telpa teisingai, perpieštas ornamentas;

c) padaryta viena klaida už mažiausio stačiakampio, kuriame telpa teisingai, perpieštas ornamentas, ribų;

d) padaryta daugiau nei viena klaida – ornamentas visai nesimetriškas arba simetrijos ašis eina per langelių vidurį ir t. t.;

Visi šie atvejai iliustruojami paveikslėliuose (39.2.a pav., 39.2.b.1-2 pav., 
39.2.c.1-2 pav., 39.2.d pav.).

39.2.a pav., 39.2.b.1-2 pav., 39.2.c.1-2 pav., 39.2.d pav.

Galima parašyti įvairiausių algoritmų, sprendžiančių šį uždavinį. Svarbu numatyti aukščiau išvardintus atvejus. 

Mūsų siūlomas algoritmas yra toks. Randame mažiausio stačiakampio, į kurį telpa perpieštas ornamentas ribas. Peržiūrime kiekvieną stačiakampio langelį. Randame du jam simetriškus langelius (vieną horizontalios, kitą – vertikalios ašies atžvilgiu). Jei nagrinėjamo langelio spalva nesutampa su vieno ar abiejų simetriškų langelių spalva, tuomet pakeičiama nagrinėjamo langelio spalva ir tikrinama, ar pakeistas ornamentas yra simetriškas. Tikrinant simetriškumą iš naujo ieškomos stačiakampio ribos.

Jei pakeistasis ornamentas nėra simetriškas, įsimenama, kad klaidos nepavyko ištaisyti, pakeistasis langelis atstatomas į pradinę padėtį ir toliau eilės tvarka peržiūrimi langeliai. Gali būti, kad šią klaidą pavyks atstatyti pakeitus kito langelio spalvą. Jei klaidą pavyko ištaisyti, darbas baigiamas. 

Trumpai sprendimo esmę galima nusakyti taip. Peržiūrimi visi langeliai, radus įtartiną langelį, keičiama jo spalva. Jei pakeitus spalvą ornamentą pavyko atstatyti darbas baigiamas, jei ne – įsimenama, kad yra klaida ir toliau peržiūrimi langeliai.

const MAX_X = 50;

      MAX_Y = 50;

type ornam = array [1..MAX_Y] of string [MAX_X];

     atsakymas = (teis, klaid, papild, pašal);

var o: ornam; { perpieštas ornamentas }
procedure pakeisk (y, x: integer);

begin
  if o[y, x]='1' then o[y, x]:= '0'

                 else o[y, x]:= '1';

end; { pakeisk }
procedure tikrink (nerasta: boolean; { ar nerasta klaida }
                   var rez: atsakymas; { atsakymo tipas }
                   var x_rez, y_rez: integer { langelio koordinatės });

{ rekursinė procedūra, kuri tikrina, ar teisingai perpieštas ornamentas; }

{radus klaidą, bandoma ją taisyti (procedūra kreipiasi į save su parametru nerasta = false). }
{ kai procedūros parametras }
{ nerasta = false, tai jokių pakeitimų ornamente daryti negalima; }
{ nerasta = true,  tai galime pakeisti vieną langelį. }
{ vartojamas globalus kintamasis o }
  var x_min, x_max, y_min, y_max,  { ribinės piešinio koordinatės }
      x, y: integer;

      ats: atsakymas;  { pagalbiniai kintamieji }
begin
  x_min := MAX_X + 1;          { randame x_min ir x_max }
  x_max := 0;

  for y := 1 to MAX_Y do
    for x:= 1 to MAX_X do
      if o[y, x] = '1' then
         begin
          if x < x_min then x_min := x;

          if x > x_max then x_max := x

         end;

  y_min := MAX_Y + 1;             { randame y_min ir y_max }
  y_max := 0;

  for y := 1 to MAX_Y do
    if pos('1', o[y]) > 0 then
       begin
         if y < y_min then y_min := y;

         if y > y_max then y_max := y

       end;

  { tikrinamas simetriškumas }
  ats:= teis;

  y := y_min;

  repeat
    x := x_min;

    repeat
      if (o[y, x] <> o[y, x_max+x_min-x]) or
         (o[y, x] <> o[y_max+y_min-y, x]) then
         if nerasta then
              begin
                pakeisk (y, x); { pataisome piešinį }
                { tikriname, ar pataisytas ornamentas }
                tikrink (false, rez, x_rez, y_rez);

                if rez = teis

                   then begin
                         x_rez:= x;

                         y_rez:= y;

                         if o[y, x] = '0'

                         then ats:= pašal { reikėjo pašalinti }
                         else ats:= papild {reikėjo papildyti}
                        end
                   else ats := klaid;

                pakeisk (y, x); { atstatome piešinį }
              end
         else ats:= klaid;

      x := x + 1;

    until (ats = klaid) and not nerasta or
          { pataisytas piešinys nesimetriškas }
          (ats in [pašal, papild]) or     { pavyko pataisyti }
          (x > xmax);             { baigėsi tikrinama eilutė }
    y := y + 1

  until (ats = klaid) and not nerasta or
        (ats in [pašal, papild]) or
        (y > ymax);               { peržiūrėtas visas piešinys }
  if ats = teis { ar simetrijos ašys eina langelių kraštinėmis }
     then if ((y_min - y_max) mod 2 <> 0) and 

             ((x_max - x_min) mod 2 <> 0)

             then rez := teis

             else rez := klaid

     else rez := ats;

end; { tikrinti }
	Testo nr.
	Komentarai pradiniams duomenis
	Rezultatai

	1
	Ornamentas yra lapo centre, nesimetriškas įstrižainių atžvilgiu
	PERPIEŠTA TEISINGAI

	2
	Ornamentas nėra lapo centre, nėra simetriškas įstrižainių atžvilgiu
	REIKIA PAPILDYTI LANGELIU (8, 46)

	3
	Ornamentas nėra lapo centre, papildomas langelis labai nutolęs nuo ornamento ir padidina eilučių skaičių
	REIKIA PAŠALINTI LANGELĮ (49, 47)

	4
	Ornamentas nėra lapo centre, yra simetriškas abiejų ašių atžvilgiu, tačiau vertikali ašis eina langelių centrais, o ne langelių kraštinėmis
	YRA DAUGIAU KAIP VIENA KLAIDA

	5
	Ornamentas nėra lapo centre, papildomas langelis sujungtas su ornamentu, dėl to stulpelių skaičius padidėja vienetu
	REIKIA PAŠALINTI LANGELĮ (6, 4)


III etapas

40. Kokoso riešutai. Šis uždavinys gali turėti ne vieną sprendinį. Tai priklauso nuo to, po kiek riešutų išsidalijo keleiviai ryte. Iš pažiūros sprendimas atrodytų akivaizdus: tarkime, kad jie ryte išsidalijo po 1 riešutą. Tuomet paskutinis keleivis paliko n riešutų. Suskaičiuokime kiek riešutų jis rado.

Jei n riešutų sudaro (n-1)-ą dalį visų rastų riešutų, tai jis rado n ( n /(n - 1) + 1 riešutą. Tačiau n ( n be liekanos nesidalija iš n - 1, taigi keleiviai ryte negalėjo išsidalyti po vieną riešutą.

Ieškokime kitokio sprendimo. Tarkime, kad pradžioje buvo m riešutų. Pirmasis keleivis juos padalijo į n dalių ir vienas liko. Taigi m galima užrašyti, kaip m = kn + 1, čia k – koks nors sveikasis teigiamas skaičius. Ieškodami mažiausio sprendinio, peržiūrėsime visas k reikšmes, pradedant nuo 1. 

function surinkta (n: integer): integer;

  var gerai,

      rasta : boolean; { ar rastas sprendinys }
      m, rieš,      { surinktų riešutų skaičius }
      i,             { bendrakeleivio numeris }
      k : integer;

begin
  rasta := false;

  k := 0;

  while not rasta do { perrinksime k variantus }
    begin
      k := k + 1;

      m := k * n + 1;

      rieš := m;

      gerai := true;

      i := 1;

      while gerai and (i <= n) do
      { tikrinama, ar tinka parinktas variantas }
        begin
          if (m - 1) mod n <> 0

             then gerai := false

             else m := m - (m - 1) div n - 1;

          i := i + 1;

        end;

      rasta := gerai and (m > 0) and (m mod n = 0)

    end;

  surinkta := rieš

end; { surinkta }
	Testo nr.
	Pradinis duomuo
	Rezultatas

	1
	2
	11

	2
	4
	765

	3
	5
	3121


Pastaba. Didesnių pradinių duomenų negalima parinkti. Jei žmonių skaičius didesnis už 5, surinktų riešutų skaičius viršija maxint.

41. Valiutos keitimas. Valiutų pardavimo lentelę galima vaizduoti orientuotu soriniu grafu. Valstybės bus grafo viršūnės, valiutos pirkimas – grafo lankai. Sąlygoje pateiktą antrą pavyzdį galėtume pavaizduoti tokiu grafu (41.1 pav.):

41.1 pav.

Kelio ilgis apibrėžiamas neįprastai – jis lygus lankų svorių sandaugai. Uždavinio tikslas: rasti bet kokį ciklą grafe, kurio ilgis yra didesnis už vienetą. 

Uždavinį sprendžiame grįžimo metodu. Pradedame nuo pirmosios valstybės. Tarkime, kad turime vieną pirmosios valstybės valiutos vienetą. Iš jos bandome keliauti į visas valstybes, perkančias pirmosios valstybės valiutą. Įsidėmime kuriose valstybėse esame buvę ir kiek pinigų turime būdai kiekvienoje valstybėje. Jei atvykstame į valstybę, kurioje jau esame buvę ir matome, kad turime daugiau pinigų nei jų turėjome ten atvykę pirmą kartą – maršrutas rastas. Beje, ši valstybė nebūtinai turi būti ta pati, nuo kurios pradėjome keliauti. Vėl imkime antrąjį sąlygos pavyzdį. Tarkime, kad pradedame keliauti nuo 4 valstybės (41.2 pav):

41.2 pav.

Matome, kad optimalus maršrutas yra: 1, 3, 2, 1.

Deja, šis algoritmas, kaip ir kiti grįžimo metodą naudojantys algoritmai, veikia gan lėtai. Galima stengiamės, kaip nors „apkarpyti“ perrinkimą. Tačiau įmanoma sudaryti tokį testą, kad „apkarpymai“ nepadės. 

Grafas nebūtinai yra jungus. Pavyzdžiui, gali būti du ar daugiau valstybių blokų, tokių, kad valstybės perka kitos valstybės valiutą tik tada, jei antroji valstybė priklauso tam pačiam blokui. Algoritme sukuriama aibė pasiekta. Atvykus į valstybę, įsidėmima, kad joje jau būta. Apvaikščiojus visas valstybes, kurios yra pasiekiamos iš pirmosios ir nesuradus tinkamo maršruto, bandoma ieškoti valstybių, neįeinančių į aibę pasiekta, t. y. priklausančių kitam blokui.

const MAX_N = 10;  { maksimalus valstybių skaičius }
type kursai = array [1..MAX_N, 1..MAX_N] of real;

     maršrutas = array [1..MAX_N] of 

                   record
                     pin: real;  { turimų pinigų kiekis }
                     v: integer  { į kurią valstybę }
                   end;          { einama }
     aibė = set of 1..MAX_N; { aibė valstybių, kuriose apsilankyta }
procedure ieškoti (n: integer; k: kursai;

                   var rasta: boolean; { ar egzistuoja maršrutas }
                   var m: maršrutas;

                   var pradžia: integer; 

                     { kurioje valstybėje prasideda maršrutas }
                   var pelnas: real);

  var pasiekta: aibė; { aplankytų valstybių aibė }
  procedure eiti (var i: integer);

  { rekursinė procedūra, ieškanti maršruto }
  { ji naudoja globalius kintamuosius: }
  { pasiekta, k, rasta, m, pradžia, pelnas }
    var j: integer;

  begin
    pasiekta := pasiekta + [i];

    for j := 1 to n do
      if (k[i, j] <> 0) and not rasta 

      { jei galima eiti į j-tąją valstybę }
      then if m[j].v = 0 { ir joje dar nebūta }
           then 

            begin
              { apskaičiuojame turimą pinigų kiekį }
              m[j].pin := m[i].pin * k[i, j];

              m[i].v := j;

              eiti (j);

              if not rasta { grįžtame per žingsnį atgal }
                 then begin
                        m[j].pin := 0;

                        m[i].v := 0

                      end;

            end
           else if m[j].pin < m[i].pin * k[i, j]

                then begin  { jei j-toje valstybėje jau būta }
                       rasta := true;

                       pradžia := j;

                       pelnas := 

                         m[i].pin*k[i,j]/m[j].pin*100-100

                     end;

  end; { eiti }
  var i, j: integer;

begin { ieškoti }
   rasta := false;

   for j := 1 to n do
     begin
       m[j].pin := 0;

       m[j].v := 0

     end;

   m[1].pin := 1.0;

   pradžia := 1; { pardėsime keliauti nuo pirmosios valstybės }
   pasiekta := [];

   for i := 1 to n do { bandome keliauti iš kiekvienos valstybės }
     if not (i in pasiekta) and not rasta

        then begin
               pradžia := i;

               m[i].pin := 1.0;

               eiti (i)

             end;

end;

Norint išspausdinti maršrutą, reikia pradėti nuo valstybės pradžia ir keliauti į valstybę m[pradžia].v ir t.t., kol pasiekiama pradinė viršūnė. Spausdinimas nėra sunkus ir problemų neturėtų kelti. Jei vis dėlto kiltų sunkumų, galima pasižiūrėti programoje, pateiktoje kartu su testais.

Galimas ir kitoks, ekonomiškesnis sprendimo variantas – ieškoti trumpiausio kelio grafe nuo kiekvienos viršūnės iki visų likusių (nes nežinia iš kurios valstybės teks pradėti keliauti). Tą atlieka Floido-Voršalo (Floyd-Warshall) algoritmas. Aklai taikyti šio algoritmo nepavyks. Mat reikia rasti ne trumpiausią ciklą, o ciklą, kurio ilgis, didesnis už vienetą. Taigi, jei dvi viršūnes jungiančio lanko svoris yra s, tai atstumu tarp dviejų viršūnių reiktų laikyti 1/s. Kuo didesni bus briaunų svoriai, tuo bus trumpesnis kelias. Iš gautų trumpiausių kelių tarp viršūnių porų reikės sukonstruoti ciklus ir iš jų išrinkti trumpiausio ilgio ciklą. 

	Testo nr.
	Paaiškinimai
	Rezultatai

	1
	Dvi valstybės
	PELNINGO MARŠRUTO NĖRA

	2
	Dvi valstybės
	PELNINGAS MARŠRUTAS: 1 2

PELNAS: 5.0%

	3
	Penkios valstybės (sudėtingesnis testas)
	PELNINGAS MARŠRUTAS: 1 2 5 3

PELNAS: 38.6%

	4
	Devynios valstybės, grafas nejungus
	PELNINGAS MARŠRUTAS: 7 8 9

PELNAS: 56.0%

	5
	Dešimt valstybių (šiuo testu tikrinamas programos efektyvumas)
	PELNINGO MARŠRUTO NĖRA


42. Algoritmų analizė. Analizei pateiktos programos yra realūs moksleivių sprendimai, sukurti olimpiados metu. Kiekvieną sprendimą analizuosime dviem aspektais: įvertinsime algoritmavimo kultūrą bei algoritmo teisingumą. 

Algoritmavimo kultūros įvertinimas

Pirmoji programa
Vienas stambesnių programos trūkumų yra tai, kad algoritmas nėra išskaidytas į procedūras ar funkcijas. Visi veiksmai atliekami pagrindinėje programos dalyje. Neaprašytas nei vienas duomenų tipas, nei viena konstanta. Dėl to sudėtinga testuoti programą. Jei derinimo metu norėtume pradiniams duomenims paimti mažesnį ornamentą, reiktų daugelyje vietų keisti programą. Nededami tarpai prieš ir po priskyrimo ženklą, aritmetinius veiksmus. Šiek tiek šlubuoja kintamųjų pavadinimai. Dalis jų greičiausiai yra angliški: h – height (angl. aukštis), w – witdh (ang. plotis). Kintamasis mas turbūt reiškia masyvą. Bet tai akivaizdu pasižiūrėjus į kintamojo tipą!

Programos tekstą autorius dėsto kiek kitaip, nei įprasta, tačiau stilius yra neblogas ir visiškai priimtinas. Kitas šios programos privalumas – aiškūs ir tikslūs komentarai. Iš vienos pusės jų nėra per daug, iš kitos – jų pakanka, norint suprasti kaip veikia algoritmas.

Antroji programa
Iš pirmo žvilgsnio į akis krinta tuščios eilutės. Programoje jų tikrai per daug. Tarpų daugiau nei pirmojoje programoje, tačiau vis vien trūksta. Yra netinkamų vardų. Pavyzdžiui programa pavadinta Olimpiada2. Iš vienos pusės šis uždavinys buvo pateiktas tikrai ne antrojoje olimpiadoje. Iš kitos pusės kiekvienoje olimpiadoje yra trys etapai ir kiekviename etape yra po kelis uždavinius. Nepagrįstai vartojamas byte tipas.

Rašant kintamųjų vardus dažnai vartojamos ir didžiosios ir mažosios raidės. Toks stilius yra vartojamas pasaulyje ir priimtinas. Deja, programos autorius jo iki galo nesilaiko.

Pats tekstas yra prasčiau išdėstytas, nei pirmojoje programoje. Ilgą rekursinę funkciją sunkoka skaityti. Nesuprantama kodėl pasirinkta funkcija o ne procedūra. Mat funkcija gauna tris rezultatus, o ne vieną.

Teigiamos savybės yra tai, kad aprašyti duomenų tipai, konstantos, programa išskaidyta į funkcijas ir procedūras, tinkamai pakomentuota. 

Trečioji programa

Palyginus su dviem pirmosiomis programomis, jos testas išdėstytas aiškiausiai ir suprantamiausiai. Tinkamai parinktas duomenų tipas ornamentui. Blogai, kad tekste vartojamas ornamento dydis, o ne konstanta. Sunkiau derinti, jei norima dirbti su mažesniu ornamentu. Trečioji programa prasčiau komentuota, nei pirmoji ar antroji.

Algoritmo skaidymas į procedūras yra labai reikalingas ir prasmingas, tačiau nesuprantama kodėl rezultatai spausdinami toje pačioje procedūroje, kur atliekami visi skaičiavimai, o ne pagrindinėje programos dalyje. Trūksta komentarų, ką žymi kintamasis m ir ką atlieka ilgasis sąlyginis sakinys case. Nekaip atrodo procedūra tikrinti be jokių parametrų, neišvardinti globalieji kintamieji. Procedūroje tikrinti nelabai prasmingai parinkti kai kurių kintamųjų vardai.

Visos trys programos turi panašių trūkumų ir visų trijų algoritmavimo kultūrą būtina taisyti. Iš kitos pusės nei viena jų nėra tiek supainiota, kad būtų visiškai nesuprantama ir paprasčiau būtų išspręsti uždavinį iš naujo, nei bandyti taisyti.

Sprendimų idėjų įvertinimas

Sprendimo idėjos vertinamos ne tik nagrinėjant programą, bet ir ją atliekant su įvairiais testais. Kiekvienas testas tikrina programos teisingumą tam tikrose situacijose. Tinkamai parinkus testus dažnai galima pasakyti kokie atvejai nenumatyti programoje, ar net apytikriaiiš viso pasakyti kokia yra sprendimo idėja.
Pirmojo sprendimo idėja

Surandamos mažiausio stačiakampio, į kurį telpa perpieštas ornamentas ribos. Apskaičiuojamas stačiakampio aukštis, plotis ir centras. Tuomet stačiakampis padalijamas į keturias dalis, imama viena dalis ir kiekvienam jos langeliui randami trys simetriški langeliai kitose dalyse. Pirmoji rasta klaida taisoma, likusios – nebe.

Sprendimo autorius nepastebėjo, kad perpiešiant ornamentą papildomas langelis gali būti užtušuotas toli nuo ornamento ir dėl to padidės mažiausias stačiakampis į kurį telpa teisingai perpieštas ornamentas.

Antrojo sprendimo idėja

Randamos mažiausio stačiakampio, į kurį telpa perpieštas ornamentas ribos. Peržiūrimas kiekvienas stačiakampio langelis. Randami du jam simetriški langeliai (vienas horizontalios ašies, kitas – vertikalios ašies atžvilgiu). Jei nagrinėjamo langelio spalva nesutampa su vieno ar abiejų simetriškų langelių spalva, tuomet pakeičiama nagrinėjamo langelio spalva ir tikrinama, ar pakeistas ornamentas yra simetriškas. Tikrinant simetriškumą iš naujo ieškomos stačiakampio ribos.

Jei pakeistasis ornamentas nėra simetriškas, įsimenama, kad klaidos nepavyko ištaisyti, pakeistasis langelis atstatomas į pradinę padėtį ir toliau eilės tvarka peržiūrimi langeliai. Gali būti, kad šią klaidą pavyks atstatyti pakeitus kito langelio spalvą. Jei klaidą pavyko ištaisyti, darbas baigiamas. 

Trumpai sprendimo esmę galima nusakyti taip. Peržiūrimi visi langeliai, radus įtartiną langelį, keičiama jo spalva. Jei pakeitus spalvą ornamentą pavyko atstatyti darbas baigiamas, jei ne – įsimenama, kad yra klaida ir toliau peržiūrimi langeliai.

Deja, sprendimo autorius užmiršo apie tai, kad simetrijos ašys eina langelių kraštais, o ne langelių viduriu. 

Trečiojo sprendimo idėja

Pirmiausia apskaičiuojamos mažiausio stačiakampio, į kurį telpa perpieštas ornamentas, centro koordinatės. Tuomet stačiakampis dalijamas į keturias dalis, imama viena dalis ir kiekvienam jos langeliui randami simetriški langeliai. Pirmą kartą klaida taisoma, antrąjį – nebe.

Kaip ir pirmuoju atveju, nepastebėta, kad perpiešiant ornamentą papildomas langelis gali būti užtušuotas toli nuo ornamento ir dėl to padidės mažiausias stačiakampis į kurį telpa teisingai perpieštas ornamentas. Beje, sprendimo autorius realizuodamas šią idėją padarė nemažai klaidų ir dėl to programa pateikia teisingus atsakymus tik nedidelei daliai testų.

Taisymui pasirinkta antroji programa. Jos stilius nemažai taisytas. Papildomos eilutės, įterptos taisant klaidą (su simetrijos ašimis) išspausdintos kursyvu.

Program PerpieštasOrnamentas;

Const
  Byla = 'ORNAM.DAT';

  MaxX = 50;

  MaxY = 50;

Type Ornam = Array [1..MaxY] of String [MaxX];

     Atsakymas = (Teis, Klaid, Papild, Pašal);

Var O : Ornam; { perpieštas ornamentas }
  Procedure Pakeisk (y, x : Integer);

  Begin
    If o[y, x]='1' Then o[y, x]:= '0'

                   Else o[y, x]:= '1';

  End; { Pakeisk }
  Procedure Tikrink (Nerasta : Boolean; { ar nerasta klaida }
                     Var Rez : Atsakymas; { atsakymo tipas }
                     Var Xrez, 

                         Yrez : integer { langelio koordinatės });

  { Rekursinė procedūra, kuri tikrina, ar teisingai perpieštas.
   ornamentas. Radus klaidą, bandoma ją taisyti (procedūra kreipiasi
   į save su parametru Nerasta = False).
   Kai procedūros parametras
   Nerasta = False, tai jokių pakeitimų ornamente daryti negalima;
   Nerasta = True,  tai galime pakeisti vieną langelį.
   Vartojamas globalus kintamasis O }
  Var xmin, xmax, ymin, ymax,  { ribinės piešinio koordinatės }
      x, y       : Integer;

      Ats        : Atsakymas;  { pagalbiniai kintamieji }
  Begin
    xmin := MaxX + 1;          { randame xmin ir xmax }
    xmax := 0;

    For y := 1 to MaxY do
      For x:= 1 to MaxX do
        If O[y, x] = '1' Then
           Begin
            If x < xmin Then xmin := x;

            If x > xmax Then xmax := x

           End;

    ymin := MaxY + 1;             { randame ymin ir ymax }
    ymax := 0;

    For y := 1 to MaxY do
      If Pos('1', O[y]) > 0 Then
         Begin
           If y < ymin Then ymin := y;

           If y > ymax Then ymax := y

         End;

    { tikrinamas simetriškumas }
    Ats:= Teis;

    y := ymin;

    Repeat
      x := xmin;

      Repeat
        If (O[y, x] <> O[y, xmax+xmin-x]) or
           (O[y, x] <> O[ymax+ymin-y, x]) Then
           If Nerasta Then
                Begin
                  Pakeisk (y, x); { pataisome piešinį }
                  { tikriname, ar pataisytas ornamentas }
                  Tikrink (False, Rez, Xrez, Yrez);

                  If Rez = Teis

                     Then Begin
                           Xrez:= x;

                           Yrez:= y;

                           If o[y, x] = '0'

                           Then Ats:= Pašal { reikėjo pašalinti }
                           Else Ats:= Papild {reikėjo papildyti}
                          End
                     Else Ats := Klaid;

                  Pakeisk (y, x); { atstatome piešinį }
                End
           Else Ats:= Klaid;

        x := x + 1;

      Until (Ats = Klaid) and not Nerasta or
            { pataisytas piešinys nesimetriškas }
            (Ats in [Pasal, Papild]) or     { pavyko pataisyti }
            (x > Xmax);             { baigėsi tikrinama eilutė }
      y := y + 1

    Until (Ats = Klaid) and not Nerasta or
          (Ats in [Pasal, Papild]) or
          (y > ymax);               { peržiūrėtas visas piešinys }
    if Ats = Teis

       Then If ((ymin - ymax) mod 2 <> 0) and 

               ((xmax - xmin) mod 2 <> 0)

               Then Rez := Teis

               Else Rez := Klaid

       Else Rez := Ats;

   end; { Tikrinti }
Var Fl    : Text;

    i,

    Xrez,Yrez : Integer;  { vietos, kurią reikia pataisyti, koordinatės }
    Rez: Atsakymas;

Begin
  Assign (Fl, Byla);

  Reset (Fl);

  For i := 1 to MaxY do
    ReadLn (Fl, O[i]);

  Close (Fl);

  Xrez:= 0; Yrez:=0;

  Tikrink (True, Rez, Xrez, Yrez);

  Case Rez of
    Teis   : WriteLn('Perpiešta teisingai.');

    Papild : WriteLn('Reikia papildyti langeliu (',Xrez,', 

                     ',Yrez,').');

    Pasal  : WriteLn('Reikia pašalinti langelį (',Xrez,', 

                     ',Yrez,').');

    Klaid  : WriteLn('Yra daugiau kaip viena klaida.')

  end;

  WriteLn;

  WriteLn ('Paspauskite Enter klavišą ...');

  ReadLn

End.

Ši programa pateikta kaip 39-tojo uždavinio sprendimas (tik jos teksto išdėstymas priderintas prie bendro knygos stiliaus).

43. Investicinių čekių aukcionas. Pradėsime nuo pirmosios dalies: pradinių duomenų skaitymo. Kainos gali kisti nuo 0.01 iki 5.00 litų, vadinasi kainų yra ne daugiau kaip 500. Uždavinio sprendimui nėra svarbu, kiek čekių ir už kokią kainą aukciono dalyvis siūlo ar perka. Svarbu: kiek iš viso čekių ir už kokią kainą parduodami ir perkami. Todėl fiksuojame kiekvieną kainą ir nuskaitydami pradinius duomenis iš karto skaičiuojame kiek čekių yra perkama ir parduodama už kiekvieną iš 500 kainų. Procedūrą, įvedančią pradinius duomenis, pateikiame kartu su pagrindine programos dalimi.
program investicinių_čekių_aukcionas;

  const MAX_KAINA = 500;   { maksimali kaina, padauginta iš 100 }
  type lentelė = array [1..MAX_KAINA] of 

                      record
                         pirk, pard: longint;

                      end;

  procedure prad_duom_skaitymas (var čekiai: lentelė);

    var f: text;

        c: char; { 'A' – pardavėjas, 'B' – pirkėjas }
        r: real; { čekių kaina }
        i, kaina, kiek: longint;

  begin
    for i := 1 to MAX_KAINA do
      begin
        čekiai[i].pard := 0; { dar neradome pirkėjų norinčių pirkti }
        čekiai[i].pirk := 0; { ar parduoti čekius už kainą i }
        end;

    assign (f, 'AUKCION.DAT');

    reset (f);

    readln (f, c, kiek, r);

    repeat
      kaina := round (r * 100);

      if c = 'A'

      then 

        čekiai[kaina].pard := čekiai[kaina].pard + kiek

      else 

        čekiai[kaina].pirk := čekiai[kaina].pirk + kiek;

      readln (f, c, kiek, r);

    until C = 'Z';

    close (f)

  end; { prad_duom_skaitymas }
var n: longint;

    čekiai: lentelė;

    kaina: real;

    kiekis: longint;

begin
  prad_duom_skaitymas (čekiai);

  histograma (čekiai);

  aukcionas (čekiai, kaina, kiekis);

  write ('UŽ ', kaina: 0: 2, ' LT. KAINĄ BUS PARDU0TA ');

  write (kiekis, ' ČEKIŲ');

end.

Histogramą nupiešti nėra sunku. Jai skirsime 20 ekrano eilučių ir 50 stulpelių. Kainų yra 500, todėl kiekvienai kainai negalime skirti atskiros eilutės. Tam randame mažiausią bei didžiausią kainas ir šį intervalą padalijame į 20 mažesnių intervalų. Kiekvienam mažesniam intervalui skiriame po vieną eilutę. 

Taip pat darome ir su čekių kiekiais. Peržiūrime visus 20 kainų intervalų ir išrenkame tą, kuriame parduodamų (ar perkamų) čekių kiekis yra didžiausias. Šį čekių kiekį padalijame iš 50. Pavyzdžiui, jei didžiausias čekių kiekis yra 10000, tai vienas simbolis atitiks 200 čekių. Gali būti taip, kad kuriame nors kainų intervale perkamų ar parduodamų čekių kiekis mažesnis nei vieną simbolį atitinkantis čekių kiekis. Susitariame labai mažam čekių kiekiui skirti vieną simbolį.

Sąlygoje pateikto pavyzdžio histograma atrodo taip (43.1 pav.):

43.1 pav.

procedure histograma (čekiai: lentelė);

{ histogramos piešimas }
 const  EILUT = 19;    { histogramos parametrai }
        STULP = 50;

  var  vi, ap,         { kainų intervalas, kuriame galima prekyba }
       koef1, koef2,   { mastelių koeficientai }
       kiek_pa, kiek_pe, { kiek parduos ir kiek pirks čekių }
       min_k, max_k,    { minimali ir maksimali kainos }
       par, per,       { parduodamų ir perkamų čekių kiekiai }
       i, j: longint;

       max: longint;

begin
  { viršutinė eilutė }
  writeln;

  write   ('  KAINA:    KIEKIS ');

  write   ('(''+'' pasiūla; ''-'' paklausa)');

  writeln ('                PASIŪLA  PAKLAUSA');

  { apskaičiuosime pasiūlymų viršutinį ir apatinį rėžius }
  ap := 1;

  while (čekiai[ap].pard = 0) and 

        (čekiai[ap].perk = 0) do
    ap := ap + 1;

  vi := MAX_KAINA;

  while (čekiai[vi].pard = 0) and 

        (čekiai[vi].perk = 0) do
    vi := vi - 1;

  { apskaičiuojamas kainų mastelis koef1 }
  koef1 := (vi - ap + 1) div EILUT;

  { jei kainų skirtumas mažesnis už ekrano eilučių skaičių }
  if koef1 = 0 then koef1 := 1;

  { apskaičiuojamas čekių kiekio mastelis koef2 }
  max := 0;

  for j := 0 to EILUT-1 do
    begin
      { kainų ribos, kurą atitinka eilutė j }
      min_k := ap + j * koef1;

      max_k := ap + (j+1) * koef1 - 1;

      { jei eilutė paskutinė – pridedame neišsidalinusį likutį }
      if j = EILUT-1 then max_k := vi;

      { rasime čekių skaičių kiekviename kainų intervale }
      { iš jų išrinksime didžiausią čekių kiekį }
      kiek_pa := 0; kiek_pe := 0;

      for i := min_k to max_k do
        begin
          kiek_pa := kiek_pa + čekiai[i].pard;

          kiek_pe := kiek_pe + čekiai[i].perk;

        end;

      if kiek_pa > max then max := kiek_pa;

      if kiek_pe > max then max := kiek_pe

    end;

  koef2 := max div STULP;

  if koef2 = 0 then koef2 := 1;

  { braižome histogramą }
  for j := 0 to EILUT-1 do
    begin
      { kainų ribos, kurą atitinka eilutė j }
      min_k := ap + j * koef1;

      max_k := ap + (j+1) * koef1 - 1;

      { jei eilutė paskutinė – pridedame neišsidalinusį likutį }
      if j = EILUT - 1 then max_k := vi;

      { spausdiname kainų ribas }
      write (min_k/100:4: 2, '-', max_k/100: 4: 2, ': ');

      { apskaičiuojame perkančiųjų ir parduodančiųjų skaičių }
      kiek_pa := 0; kiek_pe := 0;

      for i := min_k to max_k do
        begin
          kiek_pa := kiek_pa + čekiai[i].pard;

          kiek_pe := kiek_pe + čekiai[i].perk;

        end;

      { perkamų ir parduodamų čekių kiekius perskaičiuojame }
      { pagal mastelį }
      if (kiek_pa > 0) and (kiek_pa div koef2 = 0)

         then par := 1

         else par := kiek_pa div koef2;

      if (kiek_pe > 0) and (kiek_pe div koef2 = 0)

         then per := 1

         else per := kiek_pe div koef2;

      { vaizduosime čekių pirkimą ir pardavimą }
      if per >= par then { jei perka daugiau negu parduoda...}
         begin
           { rodome parduodančiųjų ir perkančiųjų skaičių (((((() }
           for i := 1 to par do
             write ('(');

           { rodome perkančiųjų skaičių   (-----) }
           for i := par+1 to per do
             write ('-');

           { ekrane matome                 ((((((-----) }
           write ('': STULP-per)

         end
       else  { jei perka mažiau negu parduoda }
           begin
             { rodome parduodančiųjų ir perkančiųjų skaičių (((((() }

             for i := 1 to per do
              write ('(');

             { rodome parduodančiųjų skaičių (+ + + +) }
             for i := per+1 to par do
               write ('+');

             { ekrane matome                 ((((((+ + + +) }
             write('': STULP-par)

         end;

      { ir užrašome tikslius skaičius }
      write (kiek_pa: 8, kiek_pe: 8);

      writeln

     end;

  { skaičiuosime kiek iš viso perkama ir parduodama čekių }
  kiek_pa := 0; kiek_pe := 0;

  for i := ap to vi do  { susumuojame }
    begin
      kiek_pa := kiek_pa + čekiai[i].pard;

      kiek_pe := kiek_pe + čekiai[i].perk;

    end;

  write ('((((((((((((((((((((((((((');

  write ('((((((((((((((');

  writeln('IŠ VISO:', '':STULP+3, kiek_pa: 8, kiek_pe: 8)

end; { histograma }
Antroji ir trečioji užduoties dalys įvykdomos lengvai. Fiksuojame kiekvieną kainą ir žiūrime kiek bus nupirkta čekių. Iš visų kainų išrenkame tą, už kurią galima parduoti daugiausia čekių. 

procedure aukcionas(čekiai: lentelė; { čekių kainos }
                  var kaina: real; { nustatyta vieno čekio kaina }
                  var kiekis: longint { parduotų čekių kiekis });

  function min (x, y: longint): longint;

  { randa mažesnįjį iš dviejų skaičių }
  begin
    if x <= y

       then min := x

       else min := y;

  end; { min }
  var čekio_kaina, i, perka, parduoda: longint;

begin
  { rasime perkamų už MAX_KAINĄ čekių kiekį }
  perka := čekiai[MAX_KAINA].pirk;

  { rasime parduodamų už MAX_KAINĄ čekių kiekį }
  parduoda := 0;

  for i := 1 to MAX_KAINA do
    parduoda := parduoda + čekiai[i].pard;

  { tokį kiekį galima parduoti už MAX_KAINĄ }
  kiekis := min (parduoda, perka);

  čekio_kaina := MAX_KAINA;

  for i := MAX_KAINA - 1 downto 1 do
    { kainą nuleidinėsime }
    begin
      perka := perka + čekiai[i].pirk;

      parduoda := parduoda - čekiai[i].pard;

      if kiekis < min (perka, parduoda)

         then begin
                kiekis := min (perka, parduoda);

                čekio_kaina := i;

              end;

    end;

  kaina := čekio_kaina/100;

end; { aukcionas }
	Testo nr.
	Parduodamų čekių skaičius
	Perkamų čekių skaičius
	Nustatyta čekio kaina
	Parduotų čekių skaičius
	Paaiškinimai

	0
	20000
	69000
	0.74 Lt
	20000
	Testas sutampa su sąlygoje pateiktu pavyzdžiu; nupirkti visi parduodami čekiai

	1
	37160
	30720
	2.80 Lt
	25760
	Nuperkami ne visi parduodami čekiai

	2
	69920
	71880
	3.20 Lt
	38440
	Atsitiktinis testas

	3
	20480
	23320
	3.31 Lt
	7720
	Kainų intervalas nuo 2.8 lt iki 3.81 lt. (tikrinamas histogramos vaizdumas)

	4
	20000
	69100
	0.74 Lt
	20000
	Už labai didelę kainą perkama tik 100 čekių (tikrinamas histogramos vaizdumas)


44. Paslaptingi žemynai. Sąlyga pirmiausia prašo pateikti statistiką apie perskaitytą žemėlapį. Tai padaryti nesunku:

const EIL = 16;

      ST = 48;

type žemėlapis = array [0..EIL+1, 0..ST+1] of char;

procedure tirti (ž: žemėlapis; var s, v: integer);

{ surenka statistinius duomenis apie pradinį žemėlapį }
  var i, j: integer;

begin
  s := 0;

  v := 0;

  for i := 1 to EIL do
    for j := 1 to ST do
      if ž[i, j] = 'V'

         then v := v + 1

         else s := s + 1

end; { tirti }
Kad nereikėtų grįžti prie statistikos, pateikiame procedūrą, teikiančią statistinę informaciją apie ištirtą žemėlapį:

procedure statistika (ž: žemėlapis;

                      var  k, { KALNAS }
                      p, { PUSIASALIS }
                      r, { PAKRANTĖ
                      o, { OKEANAS }
                      j, { JŪRA }
                      e: integer { EŽERAS });

{ surenka statistiką apie ištirtą žemėlapį }
  var i, ii: integer;

begin
  k := 0; p := 0; r := 0;

  o := 0; j := 0; e := 0;

  for i := 1 to EIL do
    for ii := 1 to ST do
      case ž[i, ii] of
       'K': k := k + 1;

       'P': p := p + 1;

       'R': r := r + 1;

       'O': o := o + 1;

       'J': j := j + 1;

       'E': e := e + 1

      end
end; { statistika }
Pradėkime tirti žemėlapį. Pirmiausia pastebėsime, kad sausumą ir vandenis galime tirti nepriklausomai. Nė vienas sausumos langelis negali virsti vandenų langeliu ar atvirkščiai. Kita vertus, sausumos langelio tipas gali priklausyti nuo to, ar jį supa vanduo, ar ne, tačiau tikrai nepriklauso nuo to, kokio tipo yra tas gretimas vandens langelis. Ir atvirkščiai: vandenų langelių tipai nepriklauso nuo sausumos langelio tipo. 

Imkime sausumos langelius. Kadangi joks sausumos langelis negali virsti vandens langeliu, tai akivaizdu, kad visus KALNO langelius galima pažymėti vieną kartą peržiūrėjus žemėlapį:

procedure kalnas (var ž: žemėlapis);

{ suranda kalno langelius }
  var i, j: integer;

begin
  for i := 1 to EIL do
    for j := 1 to ST do
      if (ž[i, j] = 'S') or (ž[i, j] = 'K')

      then if (ž[i-1, j] in ['S', 'K']) and 
              (ž[i+1, j] in ['S', 'K']) and
              (ž[i, j-1] in ['S', 'K']) and
              (ž[i, j+1] in ['S', 'K'])

           then ž[i, j] := 'K'

end; { kalnas }
Toliau reikia tikrinti, kurie sausumos langeliai yra PUSIASALIS. Šiuo atveju nepakanka peržiūrėti žemėlapį vieną kartą. Mat langelis gali virsti PUSIASALIO langeliu, jei tarp jo gretimų langelių yra PUSIASALIO langelių. Jei peržiūrėjus žemėlapį atsirado nors vienas naujas PUSIASALIO langelis, būtina žemėlapį peržiūrėti dar kartą (procedūra pusiasalis).

procedure pusiasalis (var ž: žemėlapis;

                      var pakeista: boolean);

{ suranda pusiasalio langelius dabartiniame žemėlapyje }
  procedure kas (lang: char; { nagrinėjamas langelis }
                 var vt, pus: integer);

  { nustato, ar duotasis langelis yra vandens, ar pusiasalio langelis }
  begin
    case lang of
     'V': vt := vt + 1;

     'P': pus := pus + 1

    end;

  end; { kas }
  var i, j, vt, pus: integer;

begin { pusiasalis }
  pakeista := false;

  for i := 1 to EIL do
    for j := 1 to ST do
      if ž[i, j] = 'S'

      then begin
             vt := 0;

             pus := 0;

             kas (ž[i-1, j], vt, pus);

             kas (ž[i+1, j], vt, pus);

             kas (ž[i, j-1], vt, pus);

             kas (ž[i, j+1], vt, pus);

             if (vt = 3) or (vt = 2) and (pus >= 1) or
                (vt = 1) and (pus >= 2)

             then begin
                    ž[i, j] := 'P';

                    pakeista := true

                  end
           end;

end; { pusiasalis }
Visi likę sausumos langeliai bus PAKRANTĖS langeliai. 

procedure pakrantė (var ž: žemėlapis);

{ suranda PAKRANTĖS langelius }
  var i, j: integer;

begin
  for i := 1 to EIL do
    for j := 1 to ST do
      if ž[i, j] = 'S'

          then ž[i, j] := 'R'

end; { pakrantė }
Nagrinėkime vandenis. Reikia nepamiršti, kad žemėlapio išorėje yra OKEANAS. Patogu žemėlapį apsupti vandens tipo langeliais. Tada ieškoti visų kitų OKEANO langelių galime rekursiškai: imame bet kurį vandens tipo langelį, esantį jau už žemėlapio ribų ir rekursiškai bandome eiti į keturias puses. Žemėlapį užteks patikrinti tik vieną kartą:

procedure okeanas (i, j: integer);

{ suranda OKEANO sritį, kuriai priklauso langelis (i, j) }

{ ž – globalus kintamasis }
begin
  ž[i, j] := 'O';

  if (i >= 1) and (ž[i - 1, j] = 'V')

     then okeanas (i - 1, j);

  if (i <= EIL) and (ž[i + 1, j] = 'V')

     then okeanas (i + 1, j);

  if (j >= 1) and (ž[i, j - 1] = 'V')

     then okeanas (i, j - 1);

  if (j <= ST) and (ž[i, j + 1] = 'V')

     then okeanas (i, j + 1)

end; { okeanas }
Ši procedūra naudoja globalųjį kintamąjį ž, todėl ji bus įrašyta kitos procedūros viduje.

Norint surasti JŪROS langelius žemėlapį taip pat gali tekti peržiūrėti keletą kartų. Čia gauname vienintelį atvejį, kai pakeitus langelio tipą, jis keičiamas antrą kartą. Mat tik OKEANO langelis gali tapti JŪROS langeliu:

procedure jūra (var ž: žemėlapis;

                var pakeista: boolean);

{ suranda JŪROS langelius dabartiniame žemėlapyje }
  procedure kas (lang: char; { nagrinėjamas langelis }
                 var o, jū, s: integer);

  { nustato, ar duotasis langelis yra OKEANO, JŪROS ar sausumos langelis }
  begin
    case lang of
     'O': o := o + 1;

     'J': jū := jū + 1;

     'K', 'P', 'R': s := s + 1

    end;

  end; { kas }
  var i, j, o, jū, s: integer;

begin { jūra }
  pakeista := false; 

  for i := 1 to EIL do
    for j := 1 to ST do
      if ž[i, j] = 'O' then
        begin
          s := 0;

          jū := 0;

          o := 0;

          kas (ž[i-1, j], o, jū, s);

          kas (ž[i+1, j], o, jū, s);

          kas (ž[i, j-1], o, jū, s);

          kas (ž[i, j+1], o, jū, s);

          if (jū >= 2) and (o <= 1) or (jū = 1) and (s >= 2) or
             (s >= 2) and (o >= 1)

          then begin
                 ž[i, j] := 'J';

                 pakeista := true

               end
        end
end; { jūra }
Suradę JŪROS langelius, galime gauti įdomią situaciją: OKEANAS tarsi atskirtas nuo viso OKEANO (44.1 pav.):

44.1 pav.

Visi likusieji vandenų langeliai yra EŽERO langeliai:

procedure ežeras (var ž: žemėlapis);

{ suranda ežero langelius }
  var i, j: integer;

begin
  for i := 1 to EIL do
    for j := 1 to ST do
      if ž[i, j] = 'V'

         then ž[i, j] := 'E'

end; { ežeras }
Belieka užrašyti pagrindinę procedūrą, tiriančią paslaptingus žemynus:

const EIL = 16;

      ST = 48;

type žemėlapis = array [0..EIL+1, 0..ST+1] of char;

procedure žemynas (var ž: žemėlapis);

{ ištiria žemyną }
  procedure okeanas (i, j: integer);

  { suranda OKEANO sritį, kuriai priklauso langelis (i, j) }
  begin

    …
  end; { okeanas }
  var pakeista: boolean;

      i, j: integer;

begin
  { ištirsime sausumos tipo langelius }
  kalnas (ž);

  pakeista := true;

  while pakeista do
    pusiasalis (ž, pakeista);

  pakrante (ž);

  { ištirsime vandenų tipo langelius }
  okeanas (0, 0);

  pakeista := true;

  while pakeista do
    jūra (ž, pakeista);

  ežeras (ž)

end; { žemynas }

